A simulation study exploring weighted Bayesian models to recover unbiased population estimates from weighted survey data.

WorldPop, University of Southampton

08 February 2021

Contents

1	Introduction	2
2	Methods	2
	2.1 Simulated Populations	2
	2.2 Simulated Survey Data	4
	2.3 Statistical Models	5
	2.4 Population Totals	9
3	Results	g
4	Discussion	10
\mathbf{C}_{0}	ontributing	11
	Suggested Citation	11
	License	11
\mathbf{R}	eferences	11
5	Tables	13
	5.1 Weighted-precision Model	13
	5.2 Weighted-likelihood Model	14
6	Figures	15
Δ	ppendix A: Supplementary Code	21

Note: This report assumes familiarity with Bayesian statistical models and notation, Stan and JAGS software, and the R statistical programming language.

1 Introduction

Statistical models used to map population estimates across the landscape require observations of population counts from a representative sample of locations to use as training data. These data usually come from household surveys in which populations are enumerated within geographically-defined survey locations. A stratified random sample is ideal for recovering unbiased estimates of the mean and variance of population densities. However, household surveys often implement a PPS sampling design (Probability Proportional to Size) in which locations with higher population densities are more likely to be included in the sample. This will result in biased estimates of average population densities for population modelling. Population-weighted sampling is intended to approximate random samples of individuals or households from sets of geographically clustered households, but it does not produce random samples of locations needed for geographical population models.

Our objectives here are to:

- 1. Demonstrate that a population-weighted sample results in biased estimates of population densities,
- 2. Demonstrate that model-based estimates of population totals for large areas are sensitive to this bias,
- 3. Explore Bayesian weighted-likelihood and weighted-precision approaches to produce unbiased parameter estimates, and
- 4. Demonstrate that weighted models can recover unbiased estimates of population densities and population totals from a population-weighted sample.

This analysis is intended as a theoretical foundation to support ongoing development of statistical models to estimate and map population sizes using weighted survey data as inputs.

2 Methods

We simulated populations by drawing population densities for each location from a distribution with known parameters. We then produced various types of samples from those populations: random, population-weighted, or a combination. Every population included one million locations and every sample included 2000 locations. We fit three types of models to these data trying to recover the known population parameters: unweighted model, weighted-precision model, and weighted-likelihood model.

All simulations were conducted using the R statistical programming environment (R Core Team 2020). Statistical models were fit using either the *RStan* R package (Stan Development Team 2019, 2020) or the *runjags* R package (Plummer 2003, Denwood 2016).

2.1 Simulated Populations

We used a log-normal distribution to represent population densities following the population model of Leasure et al (2020):

$$N_{i} \sim Poisson(D_{i}A_{i})$$

$$D_{i} \sim LogNormal(\mu_{i}, \sigma_{t,g})$$

$$\mu_{i} = \alpha_{t,g} + \sum_{k=1}^{K} \beta_{k} x_{k,i}$$

$$(1)$$

In this model, N_i was the observed population count and A_i was the observed settled area (ha) at location i. Population densities D_i were modelled as a function of settlement types t (e.g. urban/rural), geographic units g, and K geospatial covariates $x_{k,i}$. The regression parameters $\alpha_{t,g}$, β_k , and $\sigma_{t,g}$ estimated average population densities, effects of covariates, and unexplained residual variation, respectively.

The intended purpose of Eq. (1) was to estimate model parameters based on observed population data. For the purposes of the current simulation study, we reversed that logic. We provided pre-defined parameter values to generate simulated population data.

For our simulations we made a series of simplifying assumptions to this model. We assumed that every location i included one hectare of settled area (i.e. $A_i = 1$) and we ignored the Poisson variation so that $N_i = D_i$. We also ignored the effects of settlement type, geographic location, and covariates so that t, g, and $x_{k,i}$ were dropped from the model. These simplifying assumptions allowed us to isolate the effects of weighted sampling in the absence of these potentially confounding effects. While beyond the scope of the current report, relaxing these assumptions and assessing their effects should be the focus of future theoretical and empirical studies.

The simplified model used for our simulations was:


$$D_i \sim LogNormal(log(\mu), \sigma)$$
 (2)

Note: We modelled the median μ on the natural scale so that the parameter estimates were easier to interpret, but we kept σ on the log-scale to simplify the equations.

We simulated population densities (i.e. people per hectare) at one million locations by taking one million draws from this log-normal distribution. We repeated this for a range of parameter values for μ (i.e. 100, 250, 500) and σ (i.e. 0.25, 0.5, 0.75).

Following Eq. (2), a population where $\mu = 250$ and $\sigma = 0.5$ can be simulated across one million locations using the following R code:





2.2 Simulated Survey Data

We simulated three sampling designs:

- 1. Random sampling,
- 2. Population-weighted sampling, and
- 3. A combination of random and population-weighted sampling.

We always used a sample size of 2000 locations. We used the following R code to draw these samples.

2.2.1 Random Sample

The random sample was simply drawn using the sample function to draw 2000 samples without replacement from the simulated population densities:

2.2.2 Population-weighted Sample

To draw a population-weighted sample, we must first calculate sampling probabilities that are based on the population density at each location.

2.2.3 Combined Sample

Combined samples (random and weighted) were produced using several different proportions of random samples (i.e. 0.2, 0.5, 0.8). The total sample size for combined samples was always 2000.

```
# proportion random
prop <- 0.5
# select locations for weighted sample
i <- sample(x = 1:length(pop),
            size = 2e3*(1-prop),
            prob = w)
# select locations for random sample
j <- sample(x = 1:length(pop)[-i],</pre>
            size = 2e3*prop)
# weights for selected locations in weighted sample
w_i <- w[i]
# weights for selected locations in random sample
w_j \leftarrow rep(x = mean(w_i),
           times = n*prop)
# population densities at selected locations
D <- pop[ c(i,j) ]</pre>
```

Notice that we assigned equal weights to all of the random samples that were equal to the mean weight among the weighted samples. In other words, each random sample was given an equal weight in the model comparable to an average weighted sample. This balanced the influence of the random and weighted portions of the sample.

2.3 Statistical Models

We evaluated four statistical models:

- 1. Unweighted model (Stan),
- 2. Weighted-likelihood model (Stan),
- 3. Weighted-precision model (Stan), and
- 4. Weighted-precision model (JAGS).

The unweighted model was included to evaluate the bias that arises when fitting an unweighted model to population-weighted sample data. The weighted-precision and weighted-likelihood models were designed to use

sample weights to recover unbiased estimates of population parameters from a weighted sample. We developed the weighted-precision model for both Stan and JAGs to demonstrate that both implementations produced the same results and to provide example code for both. The weighted-likelihood approach required a direct adjustment to the likelihood that was not possible to implement in JAGS.

All models were run with four MCMC chains including a burnin period of 1000 iterations and an additional 1000 iterations that were retained for analysis. MCMC chains for all models achieved convergence. For JAGS models, convergence was defined as Gelman-Rubin statistics (potential scale reduction factors) that were less than 1.1 for all parameters (Gelman & Rubin 1992). For Stan models, convergence was defined as R-hat parameters less than 1.01 for all parameters (Stan Development Team 2020) (https://mc-stan.org/rstan/reference/Rhat.html).

2.3.1 Unweighted Log-normal

Our simplest model was a log-normal with no weights:

$$D_i \sim LogNormal(log(\mu), \sigma)$$
 (3)

Notice that this is identical to Eq. (2) that was used to generate our simulated populations. Our implementation used the following Stan model:

```
data{
 int<lower=0> n;
                          // sample size
                          // observed population densities
 vector<lower=0>[n] D;
parameters{
 real<lower=0> mu;
                         // median (natural)
 real<lower=0> sigma;
                         // standard deviation (log)
}
model{
 D ~ lognormal(log(mu), sigma); // likelihood
 mu ~ uniform(0, 2e3);
                          // prior for mu
  sigma ~ uniform(0, 5); // prior for sigma
}
```

2.3.2 Weighted-likelihood

The weighted-likelihood approach used the same log-normal model but implemented an adjustment to the likelihood for each sample based on the sample weights to account for the increased probability of including locations with high population densities in the weighted sample. We implemented this model in Stan:

```
model{
    // weighted likelihood
    for(i in 1:n){
        target += lognormal_lpdf( D[i] | log(mu), sigma ) / w[i];
    }
    mu ~ uniform(0, 2e3);    // prior for mu
    sigma ~ uniform(0, 5);    // prior for sigma
}
```

Note: The sampling probabilities w were defined in the section above (see Simulated Survey Data).

In this model, the likelihood for each sample is divided by its sampling probability—the probability of a location being selected for the sample out of the one million locations in the population. This adjustment to the likelihood reduces the influence on parameter estimates of locations that had higher sampling probabilities (i.e. locations with high population densities are over-represented in a population-weighted sample). If this model were used for a random sample, all of the weights would be equal and it would be equivalent to the unweighted model above (see Unweighted Log-normal).

2.3.3 Weighted-precision (Stan)

A potential alternative to the weighted-likelihood approach would be to scale the precision τ of the log-normal using the location-specific weights w_i . Precision τ is defined as the inverse of variance σ^2 :

$$\tau = \sigma^{-2}$$

$$\sigma = \tau^{-0.5}$$
(4)

For this model, we need to define an inverse sampling weight m_i that is scaled to sum to one across all samples:

$$m_i = \frac{w_i^{-1}}{\sum_{i=1}^n w_i^{-1}} \tag{5}$$

We will refer to these scaled inverse sampling weights as model weights m_i . Now we can specify a weighted-precision model as:

$$D_i \sim LogNormal(\mu_i, \tau_i^{-0.5})$$
$$\tau_i = \theta^{-2} m_i$$
 (6)

where θ^{-2} is a naive precision that does not account for the model weights m_i . Notice that the precision τ_i is location-specific (i.e. indexed by i) because it has been adjusted by the model weights m_i , and that $\tau_i^{-0.5}$ is an adjusted location-specific standard deviation. Where the model weights are relatively low, the location-specific precisions τ_i will be decreased to reduce the weight of those samples in the model. For our population-weighted sample, this reduced the weights of locations with high population densities that were over-represented in the sample.

Our goal was to recover an unbiased estimate of the standard deviation for the overall distribution of population densities among all locations in the population. So far, we have only estimated location-specific precisions τ_i that are dependent on location-specific model weights m_i . We derived the global standard deviation σ using a weighted average of the location-specific standard deviations $\tau_i^{-0.5}$:

$$\sigma = \frac{\sum_{i=1}^{n} \tau_i^{-0.5} \sqrt{m_i}}{\sum_{i=1}^{n} \sqrt{m_i}}$$
 (7)

We used $\sqrt{m_i}$ for this weighted average so that the weights are on the same scale as the standard deviations being averaged. Remember the model weights m_i were used to adjust a naive *precision* parameter and so here we want to use a square root transformed weight to calculate a weighted average of standard deviations $\tau_i^{-0.5}$.

Here we implemented the weighted-precision model in Stan:

```
data{
  int<lower=0> n;
                                 // sample size
                                 // observed counts
  vector<lower=0>[n] D;
  vector<lower=0,upper=1>[n] w; // sampling probabilities
transformed data{
  // model weights (scaled inverse sampling weights)
  vector<lower=0,upper=1>[n] m = inv(w) ./ sum(inv(w));
}
parameters{
  real<lower=0> mu;
                        // median
  real<lower=0> theta;
                        // naive standard deviation
}
transformed parameters{
  // location-specific weighted precision
  vector<lower=0>[n] tau = m * pow(theta,-2) ;
}
model{
  // likelihood with weighted precision
  D ~ lognormal( log(mu), sqrt(inv(tau)) );
  mu ~ uniform(0, 2e3); // prior median
  theta ~ uniform(0, 1); // prior naive standard deviation
}
generated quantities {
  // weighted average global sigma
  real<lower=0> sigma = sum( sqrt(inv(tau)) .* sqrt(m) ) / sum( sqrt(m));
}
```

2.3.4 Weighted-precision (JAGS)

We also implemented the weighted-precision model using JAGS software to provide an example of the coding differences and to demonstrate that they produce the same results. JAGS parameterizes the log-normal distribution using precision rather than standard deviation (as in Stan).

```
model{
```

```
# model weights (scaled inverse sampling weights)
m <- pow(w,-1) / sum(pow(w,-1))

for(i in 1:n){

    # likelihood with weighted precision
    D[i] ~ dlnorm(log(mu), tau[i])

    # location-specific weighted precision
    tau[i] <- pow(theta,-2) * m[i]
}

# prior for median
mu ~ dunif(0, 2e3)

# prior for naive standard deviation
theta ~ dunif(0, 1)

# weighted average global sigma
sigma <- sum( pow(tau,-0.5) * sqrt(m) ) / sum( sqrt(m) )
}</pre>
```

2.4 Population Totals

We used the fitted models to estimate total population sizes for the simulated populations. This was done by producing posterior predictions for the one million locations represented in the original simulated population data (see section Simulated Populations).

$$D_{i} \sim LogNormal(log(\mu), \sigma)$$

$$T = \sum_{i=1}^{1e6} D_{i}$$
(8)

The location-specific posterior predictions for population densities D_i (i.e. people per hectare) were assumed to equal the population count N_i for each location because we assumed that each location contained one hectare of settled area $A_i = 1$. We summed the location-specific posterior predictions for population counts across all locations to derive a posterior prediction for the total size T of the simulated population.

3 Results

Results presented here are based on simulated populations where $\mu=250$ and $\sigma=0.5$. For simulations that contained a combination of random and weighted samples, we used a 50/50 split. We explored other combinations of parameters and the conclusions presented here are consistent for all of the results. The supplementary files in Appendix A include the source code for conducting the simulations using any parameters and to reproduce all of the results presented here.

The unweighted model was able to recover the simulated "true" distribution of population densities from a random sample but not from a population-weighted sample (top panels of Fig. 1). All three weighted models were able to recover the "true" population distribution from a population-weighted sample and a combined sample (bottom panels of Fig. 1). Posterior predicted distributions of population densities were very similar for all of the weighted models (Fig. 2).

Some important differences were apparrent among the three weighted models when we looked at individual parameter estimates for the median μ , standard deviation σ , and mean $(\mu e^{0.5\sigma^2})$. Most importantly, the weighted-likelihood approach seemed to produce point-estimates rather than true Bayesian posterior distributions that

accounted for parameter uncertainty (Fig. 3). Although they did not adequately account for parameter uncertainty, the point-estimates appeared to be unbiased estimators (Tables 5, 6, and 7).

In contrast, the weighted-precision approaches produced full posteriors that accounted for parameter uncertainty (Fig. 4; Tables 5, 6, and 7). As expected, unweighted models that were fit to population-weighted samples significantly overestimated the median μ and the mean ($\mu e^{0.5\sigma^2}$), but surprisingly not the standard deviation σ (Fig. 4).

The bias that we saw in parameter estimates from an unweighted model fit to a population-weighted sample resulted in significant overestimation of the total population when the model predictions were applied across one million locations (scenario "wu" in Fig. 6; Table 4). The weighted models were all able to recover unbiased estimates of the total population, but the weighted-likelihood approach did not produce robust credible intervals (Fig. 5; Table 8). This was presumedly because of the previously mentioned issue with the weighted-likelihood failing to account for parameter uncertainty. The weighted-precision approaches produced unbiased estimators of total population and robust credible intervals (Fig. 6; Table 4).

4 Discussion

Results suggested that the weighted-precision model was the only approach that recovered unbiased estimates of population densities and population totals with robust credible intervals from population-weighted samples. The weighted likelihood model recovered unbiased estimates of population densities and totals from population-weighted samples, but did not produce robust credible intervals. The unweighted model recovered unbiased estimates with robust credible intervals from random samples, but produced significantly biased estimates of population densities and totals from population-weighted samples. For these reasons, we recommend the weighted-precision model for use with population-weighted samples and we have demonstrated that the Stan and JAGS implementations of this model produced the same results.

Our simulation made some simplifying assumptions to isolate effects of population-weighted sample designs on these statistical models. For the simulations, we assumed that area of settlement was equal in every location and we ignored the effects of settlement type (urban/rural), geographic location, and other geospatial covariates. Exploring the effects of these factors in a simulation framework was beyond the scope of this proof-of-concept study, but we encourage further investigation using a combination of empirical and simulation-based studies.

We defined the sampling weights as the probability of a location being selected for the sample. In a random sample, those probabilities would be equal across all locations in the population. In our simulation of population-weighted sampling, we simply defined this based on the proportion of the total population that occured in each location. In practice, the sampling weights used for PPS sample designs in household surveys are often much more complicated, involving weighted selection of survey clusters from a national sampling frame followed by weighted sampling of households within the selected clusters. This presents additional challenges for weighted models using these real-world weighted survey data (Gelman 2007). Our simulation demonstrated that the weighted models can recover unbiased parameter estimates from a population-weighted sample, but this only holds true if the available sampling weights represent the true probability of each survey cluster being selected from the national sampling frame.

Weighted-precision models were used to produce 100 m gridded population estimates for Zambia (WorldPop 2020a) and Democratic Republic of the Congo (Boo et al. 2020, WorldPop 2020b). These publications implemented a full version of the weighted-precision model that extended the core model of Leasure et al (2020) (Eq. (1)) to account for population-weighted sample data while also accounting for geographical location, settlement type (urban/rural), building footprints, and other geospatial covariates.

The weighted-likelihood approach seemed to produce point-estimates rather than true Bayesian posterior distributions that accounted for paramater uncertainty (Fig. 3). This was presumedly the result of the direct modification of the likelihood in the Stan model (see section Weighted-likelihood). This implementation of weighted-likelihood is not advised for adjusting a log-normal distribution, but it may be appropriate for discrete distributions, like Poisson, where sufficient statistics are appropriate (See "Exploiting sufficient statistics" in the "Efficiency Tuning" chapter of Stan User Guide (2019)). More investigation is needed on this topic to determine if this approach may be suitable for the one-inflated Poisson models (Leasure & Tatem 2020) developed to produce full coverage gridded

population estimates of people per household from census microdata (Minnesota Population Center 2019), which sometimes include weighted samples.

Contributing

This analysis and report were developed by Doug Leasure and Claire Dooley from the WorldPop Group at the University of Southampton with oversight and project support from Andy Tatem. Funding was provided by the Bill and Melinda Gates Foundation and the United Kingdom Foreign, Commonwealth & Development Office as part of the GRID3 project (OPP1182408, OPP1182425).

Suggested Citation

Leasure DR, Dooley CA, Tatem AJ. 2021. A simulation study exploring weighted Bayesian models to recover unbiased population estimates from weighted survey data. WorldPop, University of Southampton. doi:10.5258/SOTON/WP00706

License

You are free to redistribute this document under the terms of a Creative Commons Attribution-NoDerivatives 4.0 International (CC BY-ND 4.0) license.

References

Boo G, Darin E, Leasure D, Dooley C, Chamberlain H, Lazar A, Tatem A. 2020. Modelled gridded population estimates for the Kinshasa, Kongo-Central, Kwango, Kwilu, and Mai-Ndombe provinces in the Democratic Republic of the Congo 2018, version 2.0. WorldPop, University of Southampton. doi:10.5258/SOTON/WP00669. https://wopr.worldpop.org/?COD/population/v2.0.

Denwood MJ. 2016. runjags: An R package providing interface utilities, model templates, parallel computing methods and additional distributions for MCMC models in JAGS. *Journal of Statistical Software* 71:1–25. doi:10.18637/jss.v071.i09.

Gelman A. 2007. Struggles with survey weighting and regression modeling. Statistical Science 22:153–164. doi:10.1214/088342306000000691.

Gelman A, Rubin DB. 1992. Inference from iterative simulation using multiple sequences. $Statistical\ science\ 7:457-472.\ doi:10.1214/ss/1177011136.$

Leasure DR, Tatem AJ. 2020. A bayesian approach to produce 100 m gridded population estimates using census microdata and recent building footprints. WorldPop, University of Southampton. doi:10.5258/SOTON/WP00686.

Leasure DR, Jochem WC, Weber EM, Seaman V, Tatem AJ. 2020. National population mapping from sparse survey data: A hierarchical Bayesian modeling framework to account for uncertainty. *Proceedings of the National Academy of Sciences* 117:24173–24179. doi:10.1073/pnas.1913050117.

Minnesota Population Center. 2019. Integrated Public Use Microdata Series, International: Version 7.2. Minneapolis, MN: IPUMS.

Plummer M. 2003. JAGS: A program for analysis of Bayesian graphical models using Gibbs sampling. In: *Proceedings of the 3rd international workshop on distributed statistical computing.* Vienna, Austria., 1–10. http://mcmc-jags.sourceforge.net/.

R Core Team. 2020. R: A language and environment for statistical computing. Vienna, Austria: R Foundation

for Statistical Computing. https://www.R-project.org/.

Stan Development Team. 2019. Stan User's Guide. https://mc-stan.org/docs/2_26/stan-users-guide.

Stan Development Team. 2020. RStan: the R interface to Stan. R package version 2.19.3. http://mc-stan.org/.

WorldPop. 2020a. Bottom-up gridded population estimates for Zambia, version 1.0. WorldPop, University of Southampton. doi:10.5258/SOTON/WP00662. https://wopr.worldpop.org/?ZMB/population/v1.0.

WorldPop. 2020b. Bottom-up gridded population estimates for the Kinshasa, Kongo-Central, Kwango, Kwilu, and Mai-Ndombe provinces in the Democratic Republic of the Congo, version 1.0. WorldPop, University of Southampton. doi:10.5258/SOTON/WP00658. https://wopr.worldpop.org/?COD/population/v1.0.

5 Tables

5.1 Weighted-precision Model

Table 1: Median μ parameter estimate: Summary statistics for the posterior distributions from the unweighted and weighted-precision models.

	median	mean	lower	upper
pop	250.0	250.0	250.0	250.0
ru	254.0	254.0	248.4	259.3
wu	323.5	323.5	316.4	330.5
ww	254.7	254.7	249.4	259.9
cw	248.6	248.6	243.0	254.4

Table 2: Mean $(\mu e^{0.5\sigma^2})$ parameter: Summary statistics for the posterior distributions from the unweighted and weighted-precision models.

	median	mean	lower	upper
pop	283.3	283.3	283.3	283.3
ru	287.0	287.1	280.4	293.7
wu	365.6	365.7	357.4	374.4
ww	288.2	288.3	282.0	294.6
cw	285.0	285.0	278.5	292.0

Table 3: Standard deviation σ parameter: Summary statistics for the posterior distributions from the unweighted and weighted-precision models.

	median	mean	lower	upper
pop	0.500	0.500	0.500	0.500
ru	0.495	0.495	0.480	0.511
wu	0.495	0.495	0.480	0.510
ww	0.498	0.498	0.482	0.514
cw	0.523	0.523	0.507	0.540

Table 4: Derived population totals T: Summary statistics for the derived posterior distributions from the unweighted and weighted-precision models.

	median	mean	lower	upper
pop	283,246,502	283,246,502	283,246,502	283,246,502
ru	287,007,075	287,051,257	280,441,789	293,578,137
wu	365,627,317	365,697,914	357,366,590	374,421,369
ww	288,247,168	288,283,242	281,977,499	294,695,720
cw	285,019,319	285,030,703	278,418,950	292,024,475

See Fig. 4 for density plots of posteriors from Tables 1, 2, and 3; and see Fig. 6 for a barplot of Table 4.1

¹(pop) simulated "true" population; (ru) random sample data with an unweighted model; (wu) weighted sample data with an unweighted model; (ww) weighted sample data with a weighted-precision model; (cw) combined sample data (weighted and random) with a weighted-precision model.

5.2 Weighted-likelihood Model

Table 5: Median μ parameter: Summary statistics for the posterior distributions from the unweighted and weighted-likelihood models.

	median	mean	lower	upper
pop	250.0	250.0	250.0	250.0
ru	254.0	254.0	248.4	259.3
wu	323.5	323.5	316.4	330.5
ww	254.7	254.7	254.7	254.7
cw	248.5	248.5	248.5	248.5

Table 6: Mean $(\mu e^{0.5\sigma^2})$ parameter: Summary statistics for the posterior distributions from the unweighted and weighted-likelihood models.

	median	mean	lower	upper
pop	283.3	283.3	283.3	283.3
ru	287.0	287.1	280.4	293.7
wu	365.6	365.7	357.4	374.4
ww	286.1	286.1	286.1	286.1
cw	283.3	283.3	283.2	283.3

Table 7: Standard deviation σ parameter: Summary statistics for the posterior distributions from the unweighted and weighted-likelihood models.

	median	mean	lower	upper
pop	0.500	0.500	0.500	0.500
ru	0.495	0.495	0.480	0.511
wu	0.495	0.495	0.480	0.510
ww	0.483	0.483	0.483	0.483
cw	0.512	0.512	0.512	0.512

Table 8: Derived population totals T: Summary statistics for the derived posterior distributions from the unweighted and weighted-likelihood models.

	median	mean	lower	upper
pop	283,246,502	283,246,502	283,246,502	283,246,502
ru	287,007,075	287,051,257	280,441,789	293,578,137
wu	365,627,317	365,697,914	357,366,590	374,421,369
ww	286,114,506	286,113,499	285,826,808	286,398,643
cw	283,260,492	283,258,895	282,950,784	283,558,666

See Fig. 3 for density plots of posteriors from Tables 5, 6, and 7; and see Fig. 5 for a barplot of Table 8.2

²(pop) simulated "true" population; (ru) random sample data with an unweighted model; (wu) weighted sample data with an unweighted model; (ww) weighted sample data with a weighted-likelihood model; (cw) combined sample data (weighted and random) with a weighted-likelihood model.

6 Figures




Figure 1: Posterior predicted distributions of population densities from an unweighted model and a weighted-precision model using a random sample, a population-weighted sample, and a combination sample. Only the weighted-precision model results are shown here, but results were comparable with the weighted-likelihood model.




Figure 2: Posterior predicted distribution of population densities from weighted-likelihood, Stan weighted-precision, and JAGS weighted-precision models. The true population distribution includes one million locations and the weighted sample includes 2000 locations.




Figure 3: Parameter estimates for the median, mean, and sigma from the weighted-likelihood model. The four scenarios are: (ru) random sample data with an unweighted model; (wu) weighted sample data with an unweighted model; (ww) weighted sample data with a weighted model; and (cw) combined sample data (weighted and random) with a weighted model. The vertical red line is the simulated 'true' parameter estimate.




Figure 4: Parameter estimates for the median, mean, and sigma from the weighted-precision model. The four scenarios are: (ru) random sample data with an unweighted model; (wu) weighted sample data with an unweighted model; (ww) weighted sample data with a weighted model; and (cw) combined sample data (weighted and random) with a weighted model. The vertical red line is the simulated 'true' parameter estimate.




Figure 5: Posterior predicted population totals from a weighted-likelihood model. Population totals include one million locations from the full simulated population. The four scenarios are: (ru) random sample data with an unweighted model; (wu) weighted sample data with an unweighted model; (ww) weighted sample data with a weighted model; and (cw) combined sample data (weighted and random) with a weighted model. The horizonal line is the simulated 'true' total population size.




Figure 6: Posterior predicted population totals from a weighted-precision model. Population totals include one million locations from the full simulated population. The four scenarios are: (ru) random sample data with an unweighted model; (wu) weighted sample data with an unweighted model; (ww) weighted sample data with a weighted model; and (cw) combined sample data (weighted and random) with a weighted model. The horizonal line is the simulated 'true' total population size.

Appendix A: Supplementary Code

All code used to conduct the simulation analyses and produce results are provided in the file leasure2021simulation_supplement that is available from http://doi.org/10.5258/SOTON/WP00706. This code can be used to replicate the results in this report and to explore different simulation settings and model designs.

This zip file contains three scripts:

- 1. A minimum example of the simulation framework and weighted models.
- 2. A script to evaluate all three weighted models using the same simulated sample data.
- 3. A script to compare unweighted and weighted models using random, weighted, and combined samples.

The zip file also contains source code for four models:

- 1. Unweighted model (Stan)
- 2. Weighted-likelihood model (Stan)
- 3. Weighted-precision model (Stan)
- 4. Weighted-precision model (JAGS)